

SCIENCE & VIE

MENSUEL

N° 853 OCTOBRE 1988

130 FB - 5,20 FS - Scan 3,50 - 550 Plus - 480 Esc - 18,50 Dh - 2,300 Dt - 4000 L - USA NYC \$3,75 - OTHER \$ 3,95 - ACl - 1370 CFA

DES FEUX ROUGES DANS LE CIEL

La terre est-elle un être vivant?

M 2578 - 853 - 18,00 F



3792578018002 08530

Complétez la série

INFORMATIQUE AMUSANTE

Nous vous proposons ce mois-ci un programme de jeu des plus simples : l'Amstrad vous proposera des séries de chiffres ou de lettres et vous demandera de les compléter de manière logique ; un temps maximum de réflexion sera autorisé pour chaque problème posé ; chaque bonne réponse créditera le score d'un point (à l'inverse, toute erreur provoquera la fin de la partie). Pour obtenir le meilleur résultat, il faudra donc compléter, sans erreur et dans le délai imparti, un maximum de séries. Notons qu'il sera toutefois possible de choisir le niveau de difficulté des problèmes à résoudre et

que ce programme peut servir de base à un programme plus élaboré, c'est-à-dire comportant une plus grande diversité de cas.

Pour écrire ce programme, nous demanderons d'abord à l'Amstrad de fonctionner en mode 1, autrement dit en mode standard, le choix du niveau de jeu devant être précisé ensuite. Pour cela, les lignes 90 et 100 afficheront divers commentaires sur l'écran et la ligne 110 indiquera ce niveau. Le contrôle, à savoir la vérification de la "fourchette" sélectionnée, sera effectué par la ligne 150. Les lignes 200 à 290 rappelleront les règles du jeu. Le programme s'arrêtera en

ligne 300, les suivantes portant sur la partie elle-même, qui commencera dès qu'une touche du clavier sera cliquée.

La variable SN, permettant de déterminer le premier chiffre de chaque suite numérique, sera alors déterminée de manière aléatoire à la ligne 350 et deux tableaux seront préparés : le premier (S\$) sera chargé de la mémorisation des suites alphabétiques ; le second (S), des séries numériques (lignes 400 et 410). Chacun sera "prérempli", ceci notamment afin d'effacer les valeurs précédentes entre chaque tirage d'une nouvelle série. Pour ce qui est du chronométrage, il sera accompagné d'un signal sonore dont l'enveloppe figurera à la ligne 540.

Le remplissage des tableaux sera effectué différemment selon qu'il s'agira de séries alphabétiques

```

10 REM *****
20 REM *   CHOIX DU MODE DE FONCTIONNEMENT   *
30 REM *****
40 MODE 1:CLS
50 CLEAR
60 REM *****
70 REM *   CHOIX DU MODE DU NIVEAU DE JEU   *
80 REM *****
90 LOCATE 8,5:PRINT "choix de votre niveau;"
100 LOCATE 8,9:PRINT "5= facile; 9= difficile."
110 LOCATE 8,13:INPUT "votre choix de 5 a 9 ":F
120 REM *****
130 REM *   CONTROLE DU NIVEAU CHOISI.   *
140 REM *****
150 IF F<5 OR F>9 THEN GOTO 110
160 CLS:LET SC=0:LET RC$=""
170 REM *****
180 REM * AFFICHAGE DE LA PAGE DE PRESENTATION. *
190 REM *****
200 LOCATE 1,1:PRINT "JE VAIS VOUS PROPOSER DE COM
PLETER"
210 LOCATE 1,3:PRINT "DES SEQUENCES LOGIQUES."
220 LOCATE 1,5:PRINT "COMPORTANT SOIT UNE SUITE DE
LETTRES"
230 LOCATE 1,7:PRINT "SOIT UNE SERIE DE NUMEROS."
240 LOCATE 1,9:PRINT "VOTRE TEMPS DE REFLEXION MAX
IMUM SERA DE"
250 LOCATE 1,11:PRINT "DIX SECONDES POUR CHAQUE QU
ESTION POSEE."
260 LOCATE 1,13:PRINT "MAIS ATTENTION NE FOURNISSE
Z PAS"
270 LOCATE 1,15:PRINT "DE REPONSE ERONNEE."
280 LOCATE 1,17:PRINT "DANS CE CAS LA PARTIE SERAI
T TERMIEE."
290 LOCATE 2,24:PRINT "QUAND VOUS SEREZ PRETS TAPE
Z UNE TOUCHE"
300 IF INKEY$="" THEN GOTO 300
310 REM *****
320 REM *   DETERMINATION ALEATOIRE DU CHIFFRE   *
330 REM *   DE DEPART DE LA SERIE NUMERIQUE.   *
340 REM *****
350 LET SN=INT(RND*8)*2
360 REM *****
370 REM *   DIMENTIONNEMENT DES TABLEAUX DE   *
380 REM *   DE MEMORISATION DES SERIES.   *
390 REM *****
400 DIM S$(F+1)
410 DIM S(F+1)
420 CLS
430 REM *****
440 REM *   PREREMPLISSAGE DES TABLEAUX   *
450 REM *****
460 FOR I=1 TO F+1
470 LET S(I)=SN
480 LET S$(I)=""
490 NEXT I
500 REM *****
510 REM *   DETERMINATION DE L'ENVELOPPE SONORE   *
520 REM *   DU "DING" CHRONOMETRE.   *
530 REM *****
540 ENV 1,1,15,1,14,-1,3,1,-1,10
550 REM *****
560 REM *   REMPLISSAGE DU TABLEAU DES   *
570 REM *   SUITES DE LETTRES.   *
580 REM *****
590 FOR I=1 TO F
600 LET S$(I)=CHR$(INT(RND*26)+65)
610 NEXT I
620 REM *****
630 REM *   CHOIX DU TYPE DE SUITE NUMERIQUE   *
640 REM *****
650 LET N=INT(RND*(F-2))+2
660 LET CH=INT(RND*4)+1
670 REM *****
680 REM *   AGUILLAGE VERS LE TYPE DE SUITE CHOISI   *
690 REM *****
700 ON CH GOTO 740,820,900,980
710 REM *****
720 REM *   TYPE DE SUITE NUMERO 1   *
730 REM *****
740 FOR I=1 TO F+1
750 LET S(I)=I+N
760 NEXT I
770 LET RC=S(I-1)
780 GOTO 1070
790 REM *****
800 REM *   TYPE DE SUITE NUMERO 2   *
810 REM *****
820 FOR I=1 TO F+1
830 LET S(I)=I*N
840 NEXT I
850 LET RC=S(I-1)
860 GOTO 1070

```

ou numériques. Dans le premier cas, il se fera par tirage aléatoire de chaque caractère par la boucle FOR... NEXT des lignes 590 à 610. Dans le second, quatre types de suites différentes, donc de progressions, seront possibles :

- la multiplication par une constante ;
- l'addition d'une constante au chiffre précédent ;
- la multiplication du chiffre précédent par une constante ;
- la variable CH, enfin, tirée de manière aléatoire par la ligne 700, aiguillera le programme vers un certain type de remplissage. Celui de la suite 1, dont la mémorisation est effectuée par les lignes 740 à 780, est très simple : la progression entre deux chiffres successifs est constante et obtenue par addition d'un chiffre, au départ aléatoire, mais fixe sur toute la série.

Une fois ces séries déterminées, il faudra procéder à leur affichage. La sélection (alphabétique ou numérique) sera effectuée aléatoirement en fonction de la valeur prise par la variable A (ligne 1080). Si A prend pour valeur 0, le programme exploitera les lignes 1130 à 1240 et affichera une suite de lettres deux fois bout à bout, puis jusqu'à un caractère déterminé par la valeur de CH. Si, au contraire, A vaut 1, le programme se rendra directement en ligne 1280 et affichera la série numérique. Restent le chronométrage et le contrôle des réponses fournies. Ils seront respectivement assurés par les lignes 1350 à 1430 et 1480 à 1550 pour les séries de lettres, et par les lignes 1590 à 1640 et 1700 à 1770 pour les séries numériques.

L'utilisation de ce programme est extrêmement simple. Après sai-

sie, la commande RUN dictera à l'ordinateur de vous faire préciser votre niveau de jeu. Il suffira alors de le taper sur le clavier et de le valider par ENTER (ou RETURN suivant le type d'appareil) : un résumé des règles du jeu s'affichera sur l'écran. Comme nous l'avons vu, la partie débutera dès cliquage d'une touche quelconque du clavier. Les réponses devront être fournies pour chaque problème posé dans un délai de 10 secondes. Tant qu'elles seront exactes, et fournies dans le temps imparti, le jeu rebouclera. Dans le cas contraire, la partie sera considérée comme terminée. Un récapitulatif fournissant la réponse, ainsi que le score obtenu, sera affiché sur l'écran. Ici encore, la frappe d'une des touches du clavier permettra de relancer une nouvelle partie.

Henri-Pierre Penel

```

870 REM *****
880 REM *           TYPE DE SUITE NUMERO 3           *
890 REM *****
900 FOR I=1 TO F+1
910 IF I>1 THEN LET S(I)=S(I-1)+N
920 NEXT I
930 LET RC=S(I-1)
940 GOTO 1070
950 REM *****
960 REM *           TYPE DE SUITE NUMERO 4           *
970 REM *****
980 FOR I=1 TO F+1
990 IF I>1 THEN LET S(I)=S(I-1)*N
1000 NEXT I
1010 LET RC=S(I-1)
1020 GOTO 1070
1030 REM *****
1040 REM *   CHOIX DE L'AFFICHAGE D'UNE SUITE   *
1050 REM *           DE CHIFFRES OU DE LETTRES           *
1060 REM *****
1070 LET A=INT(RND*2)
1080 LET ALL=1
1090 IF A=1 THEN GOTO 1280
1100 REM *****
1110 REM *   AFFICHAGE D'UNE SUITE DE LETTRES.   *
1120 REM *****
1130 LET CF=INT(RND*(F-1))+1
1140 LOCATE 1,7
1150 FOR C=1 TO 2
1160 FOR I=1 TO F
1170 PRINT S$(I);
1180 NEXT I
1190 NEXT C
1200 FOR I=1 TO CF
1210 PRINT S$(I);
1220 NEXT I
1230 LOCATE 10,15:PRINT "VOTRE REPONSE ?"
1240 GOTO 1350
1250 REM *****
1260 REM *   AFFICHAGE D'UNE SUITE DE CHIFFRES   *
1270 REM *****
1280 LOCATE 1,7
1290 FOR I=1 TO F
1300 PRINT S(I);
1310 NEXT I
1320 REM *****
1330 REM *           CHRONOMETRAGE SUITE DE LETTRES.   *
1340 REM *****

```

```

1350 LET TM=10
1360 IF A=1 THEN GOTO 1590
1370 LET K$=UPPER$(INKEY$)
1380 LOCATE 1,22:PRINT "TEMPS:";TM
1390 SOUND 1,60,0,0,1,0
1400 IF K$<>" " OR TM=0 THEN GOTO 1480
1410 FOR T=1 TO 1000:NEXT T
1420 LET TM=TM-1
1430 GOTO 1370
1440 REM *****
1450 REM *           CONTROLE DE LA REPONSE ET           *
1460 REM *           REBOUCLAGE OU FIN DE PARTIE           *
1470 REM *****
1480 IF S$(CF+1)=K$ THEN LET SC=SC+1:GOTO 420
1490 SOUND 1,120,0,0,1,0
1500 CLS
1510 LOCATE 1,5:PRINT "DESOLE LA REPONSE ETAIT:";S
$(CF+1)
1520 LOCATE 1,10:PRINT "VOTRE SCORE EST DE :";SC
1530 LOCATE 1,15:PRINT "POUR UNE AUTRE PARITE TAPE
R UNE TOUCHE."
1540 IF INKEY$="" THEN GOTO 1540
1550 GOTO 40
1560 REM *****
1570 REM *           AFFICHAGE D'UNE SUITE NUMERIQUE   *
1580 REM *****
1590 LOCATE 1,15:PRINT "VOUS AVEZ 10 SECONDES DE R
EFLEXION."
1600 FOR I=10 TO 1 STEP -1
1610 LOCATE 1,22:PRINT "TEMPS:";I
1620 SOUND 1,60,0,0,1,0
1630 FOR T=1 TO 1000:NEXT T
1640 NEXT I
1650 CLS
1660 REM *****
1670 REM *           CONTROLE DE LA REPONSE ET           *
1680 REM *           REBOUCLAGE OU FIN DE PARTIE           *
1690 REM *****
1700 LOCATE 10,12:INPUT "VOTRE REPONSE ";R
1710 IF R=RC THEN LET SC=SC+1:GOTO 420
1720 CLS
1730 LOCATE 1,5:PRINT "DESOLE LA REPONSE ETAIT:";R
C
1740 LOCATE 1,10:PRINT "VOTRE SCORE EST DE :";SC
1750 LOCATE 1,15:PRINT "POUR UNE AUTRE PARITE TAPE
R UNE TOUCHE."
1760 IF INKEY$="" THEN GOTO 1760
1770 GOTO 40

```